
Fonction RECHERCHEX

Vous avez déjà voulu rechercher des valeurs avec Excel ? Laquelle des fonctions suivantes 
avez-vous utilisée ?

La fonction dynamique RECHERCHEV ? La fonction horizontale RECHERCHEH ?  
La fonction compliquée EQUIV ?

Même si les fonctions Excel ci-dessus peuvent faire l’affaire, elles ont leurs propres limites .  
Une solution à ce problème consiste à utiliser la fonction Excel introduite par Microsoft 365 : 
RECHERCHEX ! 

Si vous utilisez l’une des anciennes versions d’Excel (2010, 2013, 2016, 2019), vous ne pourrez 
pas appeler cette fonction .

La fonction RECHERCHEX fournit un remplacement polyvalent et exceptionnel aux fonctions 
Excel susmentionnées . Elle permet de rechercher rapidement des valeurs dans un ensemble de 
données (tout comme RECHERCHEV) avec plus de souplesse et des avantages supplémentaires :

• Elle peut rechercher des données à droite ou à gauche des valeurs de recherche .

• Elle recherche une correspondance exacte par défaut . Vous n’avez pas besoin d’entrer  
un argument supplémentaire pour cela .

• Elle permet de fournir une valeur ou un texte personnalisé si le résultat de la 
recherche n’est pas trouvé .

• Elle peut effectuer une recherche de correspondance partielle à l’aide de caractères 
génériques .

• Elle peut rechercher des valeurs à la fois horizontalement et verticalement .

• Elle peut renvoyer une plage de valeurs au lieu d’une valeur unique, ce qui permet 
d’étaler les résultats .

• Elle permet de trouver la dernière occurrence dans les données .

Vous êtes impatient ? Lisez la suite pour en savoir plus sur RECHERCHEX !

RECHERCHEX - Introduction
Que fait cette fonction ?

RECHERCHEX peut être utilisée pour rechercher une valeur spécifique dans un tableau et 
renvoyer la valeur de la même ligne à partir d’un autre tableau . Elle peut :

• rechercher la valeur à la fois horizontalement et verticalement ;

• effectuer une correspondance exacte ou approximative ;

• utiliser des caractères génériques ;

• renvoyer un texte personnalisé si aucun résultat n’est trouvé ;

• elle n’a même pas la restriction imposant au tableau de retour d’être à droite du tableau 
de recherche .

N’est-ce pas incroyable ? C’est une fonction puissante qui améliore les fonctions de recherche 
existantes !

Syntaxe de la formule :

=RECHERCHEX (valeur_recherchée ; tableau_recherche ; tableau_renvoyé ; 
[si_non_trouvé];[mode_correspondance];[mode_recherche])

où :

• valeur_recherchée – la valeur à rechercher ;

• tableau_recherche – plage ou tableau dans lesquels vous cherchez la valeur ;

• tableau_renvoyé – plage ou tableau dans lesquels vous cherchez le résultat ;

• [si_non_trouvé] – la valeur que vous souhaitez afficher si aucun résultat n’est trouvé ;

• [mode_correspondance]

◊ 0 : correspondance exacte (si aucun résultat trouvé, alors erreur)

◊ –1 : correspondance exacte ou inférieure suivante (si aucun résultat trouvé, la valeur 
immédiatement inférieure est affichée)

◊ 1 : correspondance exacte ou supérieure suivante (si aucun résultat trouvé, la valeur 
immédiatement supérieure est affichée)

◊ 2 : caractères génériques

• [mode_recherche] 

◊ 1 : rechercher à partir de la première valeur

◊ –1 : rechercher à partir de la dernière valeur

◊ 2 : recherche binaire ascendante

◊ –2 : recherche binaire descendante

Comment utiliser RECHERCHEX ?
L’exemple ci-dessous présente deux tableaux :

• Liste de stock contenant le code, le nom, le prix et le coût du produit

• Tableau des commandes contenant les quantités commandées

Vous voulez extraire le prix des produits depuis le tableau de la liste de stock à l’aide de 
RECHERCHEX .

ÉTAPE 1 : Entrez la fonction RECHERCHEX dans une cellule vide :

=RECHERCHEX(

ÉTAPE 2 : Saisissez le premier argument de la fonction – valeur_cherchée  
(le code du produit recherché) :

=RECHERCHEX(H10;

ÉTAPE 3 : Saisissez le deuxième argument RECHERCHEX – tableau_recherche (le tableau 
qui contient tous les codes de produit). Appuyez sur F4 après avoir sélectionné la plage de 
recherche, pour obtenir des valeurs absolues .

=RECHERCHEX(H10;$E$9:$E$12

ÉTAPE 4 : Saisissez le troisième argument RECHERCHEX - tableau_renvoyé (le tableau 
qui contient le prix de l’article cherché) . Appuyez sur F4, puis sur Entrée .

=RECHERCHEX(H10;$E$9:$E$12;$C$9:$C$12)

Comme vous pouvez le constater, Excel a extrait le prix du code LP1411 de la liste de stock et a 
fourni le résultat (650 $) dans la cellule .

Appliquez la même formule au reste des cellules en faisant glisser vers le bas la poignée 
du coin inférieur droit .

Que faire si vous recherchez un nom de produit qui n’est pas dans le tableau de la liste de stock ?

Excel vous enverra une erreur .

Pour éviter d’afficher cette erreur, vous pouvez ajouter un texte personnalisé (par exemple, 
Aucun produit trouvé) à afficher .

Pour ce faire, il suffit d’ajouter le quatrième argument optionnel de la fonction 
RECHERCHEX - [si_non_trouvé].

=RECHERCHEX(H10;$E$9:$E$12;$C$9:$C$12; "Aucun produit trouvé")

Il s’agissait d’un exemple de base de l’utilisation de RECHERCHEX . Explorons en détail  
les utilisations avancées de cette fonction .

Correspondance approximative
Dans cet exemple, Excel va rechercher le revenu donné dans la cellule E9 et trouver  
le taux d’imposition correspondant dans la colonne C . 

En plus de rechercher une correspondance exacte, Excel peut aussi rechercher une correspon-
dance approximative . Si une correspondance exacte n’est pas trouvée, il cherchera le prochain 
élément plus petit ou plus grand en fonction de l’entrée fournie .

Si le revenu est supérieur ou égal à 0 $, le taux d’imposition est de 13 % . De même, si  
le revenu est supérieur ou égal à 8 456 $, le taux d’imposition est de 18 %, et ainsi de suite .

Utilisons donc cette fonction pour déterminer le taux d’imposition pour le montant  
du revenu mentionné dans la cellule E9 .

=RECHERCHEX (valeur_cherchée ; tableau_recherche ; tableau_renvoyé ;  
               [si_non_trouvé] ; [mode_correspondance] ; [mode_recherche])

Ici, les trois arguments permanents et l’argument facultatif [mode_correspondance] sont 
utilisés . Les arguments [si_non_trouvé] et [mode_recherche] sont ignorés . Voici comment 
rechercher une correspondance approximative à l’aide de RECHERCHEX .

ÉTAPE 1 : Saisissez le premier argument valeur_cherchée – le montant du revenu 
mentionné dans la cellule E5 .

=RECHERCHEX(E5

ÉTAPE 2 : Saisissez le deuxième argument tableau_recherche – la plage contenant  
les tranches de revenus (B5:B10) .

=RECHERCHEX(E5;B5:B10

ÉTAPE 3 : Saisissez le tableau_renvoyé .

=RECHERCHEX(E5;B5:B10;C5:C10

ÉTAPE 4 : Entrez le 5e argument [mode_correspondance] . Les valeurs acceptées pour  
cet argument sont :

• 0 : correspondance exacte (si aucun résultat trouvé, alors erreur)

• –1 : correspondance exacte ou prochaine plus petite (si aucun résultat trouvé, alors  
la prochaine plus petite valeur sera affichée)

• 1 : correspondance exacte ou plus grande suivante (si aucun résultat n’est trouvé,  
la plus grande valeur suivante sera affichée)

• 2 : caractères génériques .

Dans cet exemple, la valeur sera -1 .

=RECHERCHEX(E5;B5:B10;C5:C10;;-1)

Le 4e argument est ignoré, c’est pourquoi il faut deux « ; » .

 Voici la formule à utiliser :

=RECHERCHEX(E5;B5:B10;C5:C10;;-1)

Cette fonction effectue une correspondance exacte par défaut ; il faut donc utiliser l’argument 
facultatif de la fonction, [mode_correspondance] . Ainsi, si Excel ne trouve pas une corres-
pondance exacte, il recherchera la plage de revenus immédiatement inférieure mentionnée 
dans le tableau .

L’un des avantages de l’utilisation de cette fonction par rapport à RECHERCHEV pour une 
correspondance approximative est que vous n’avez pas besoin de trier les données en ordre 
croissant . Excel s’en charge tout seul !

Cet exemple présente les données non classées en ordre croissant .

Vous pouvez utiliser la même formule dans des données classées de manière aléatoire et Excel 
fournira le même résultat correct .

272  Fonction RECHERCHEX



F
o

rm
u

le
s

Recherche horizontale
La syntaxe pour effectuer une recherche horizontale à l’aide de RECHERCHEX est la même que 
pour la recherche verticale . 

Il suffit de fournir à Excel l’objet recherché et un tableau de retour, son orientation 
n’ayant aucune importance pour la fonction RECHERCHEX .

Dans cet exemple, le nom du produit est affiché à la ligne 6 et le prix à la ligne 7 . Il faut 
effectuer une consultation horizontale pour obtenir les résultats .

ÉTAPE 1 : Saisissez la valeur de référence, soit le nom du produit mentionné dans la cellule H7 .

=RECHERCHEX(H7

ÉTAPE 2 : Saisissez le tableau de recherche, soit le tableau contenant le nom du produit . 
Appuyez sur F4 pour rendre fixes les références du tableau .

=RECHERCHEX(H7;$C$6:$F$6

ÉTAPE 3 : Saisissez le tableau de retour, soit le tableau qui doit donner les prix du produit . 
N’oubliez pas d’appuyer sur F4 .

=RECHERCHEX(H7;$C$6:$F$6;$C$7:$F$7)

C’est aussi simple que cela d’effectuer une consultation horizontale à l’aide de la nouvelle 
fonction RECHERCHEX !

Retourner une plage au lieu d’une valeur
La fonction RECHERCHEX a la possibilité de renvoyer plusieurs valeurs au lieu d’une seule en cas 
de correspondance exacte . Il ne faut pas modifier la syntaxe, il suffit d’entrer la plage entière 
dans la fonction au lieu d’une seule colonne ou ligne .

Dans cet exemple, nous voulons extraire tous les détails relatifs au nom du produit mentionné 
dans la cellule G5 .

ÉTAPE 1 : Sélectionnez la cellule contenant la valeur de recherche .

=RECHERCHEX(G5

ÉTAPE 2 : Sélectionnez la plage contenant la liste des produits et appuyez sur F4 .

=RECHERCHEX(G5;$B$4:$B$7

ÉTAPE 3 : Sélectionnez la plage entière contenant les colonnes du code, du prix et du coût . 
Appuyez sur F4 .

=RECHERCHEX(G5;$B$4:$B$7;$C$4:$E$7)

Excel va extraire ou « répandre » toutes les valeurs à l’aide de la formule RECHERCHEX .

Utilisation des caractères génériques
RECHERCHEX permet de rechercher une correspondance partielle en utilisant des caractères 
génériques tels que : * ? ~

Cette opération peut également être effectuée à l’aide de Excel RECHERCHEV .  
Mais un problème se pose lorsque vous essayez de rechercher le caractère de  
remplacement lui-même dans les données .

Avec RECHERCHEX, vous pouvez utiliser des caractères génériques dans l’argument  
valeur_cherchée et le spécifier dans l’argument [mode_correspondance] . Saisissez 2  
comme valeur mode_correspondance pour indiquer à Excel que vous essayez d’effectuer  
une correspondance partielle .

Les trois caractères spéciaux reconnus par Excel sont les suivants :

• L’astérisque (*) représente un nombre quelconque de caractères . Par exemple, Charl* 
pourrait signifier Charlie, Charlotte, Charles, etc .

• Le point d’interrogation (?) représente un caractère . Par exemple, ?oe peut signifier 
Noe, Joe, Zoe, etc .

• Le tilde (~) sert à indiquer que l’astérisque ou le point d’interrogation n’est pas  
un caractère générique . Il doit être placé juste avant * ou ?

Les exemples présentent l’utilisation des caractères génériques * ou ? pour trouver  
une correspondance partielle et renvoyer l’adresse électronique correspondante .

Char* recherche la première correspondance commençant par Char et un nombre quelconque 
de caractères après cela (Charlotte) et renvoie l’adresse électronique correspondante 
(charlotte_tan@gmail.com) .

=RECHERCHEX(E5;$B$4:$B$10;$C$4:$C$10;;2)

?oe recherche la première correspondance commençant par n’importe quel caractère suivi  
de oe (Zoe) et renvoie l’adresse électronique correspondante (zoe_velez@outlook.com) .

=RECHERCHEX(E6;$B$4:$B$10;$C$4:$C$10;;2)

Recherche de bas en haut
Cet exemple recherche une valeur du bas vers le haut au lieu de la direction par   
défaut (de haut en bas) . Pour cela, vous devez utiliser le 6e argument de la fonction,  
soit [mode_recherche] . Il s’agit d’une fonctionnalité très pratique si vous souhaitez  
trouver la dernière occurrence d’un élément dans les données . 

Fonction RECHERCHEX  277



Cet exemple fournit la date de la vente, le nom du produit et le prix mentionné . Il faut trouver 
le dernier prix auquel un produit spécifique a été vendu .

ÉTAPE 1 : Saisissez la valeur recherchée, soit le nom du produit mentionné dans la cellule F5 .

=RECHERCHEX(F5

ÉTAPE 2 : Saisissez le tableau de recherche, soit le tableau contenant le nom du produit .

=RECHERCHEX(F5;C4:C30

ÉTAPE 3 : Indiquez le tableau renvoyé, soit le tableau qui contiendra le ou les prix du produit .

=RECHERCHEX(F5;C4:C30;D4:D30

ÉTAPE 4 : Entrez le 6e argument [mode_recherche] . Ce sera –1 dans cet exemple, car  
vous cherchez de bas en haut . N’oubliez pas de saisir trois fois ; avant l’argument. 

=RECHERCHEX(F5;C4:C30;D4:D30;;;-1)

Excel commence la recherche à partir du bas et la première correspondance est extraite 
comme résultat .

